Experience from the recent HDF5 output development including OOP in FORTRAN

TSVV-5 VC

J. Gonzalez; 09-07-2021

DIFFER

NWO



HDF5 output

Structure of HDF5 output

- HDF5 is organized in datasets. A single file can contain datasets with different size and value types. Datasets can be grouped.
- Metadata can be added to the datasets to include relevant information.
- Current implementation in Eirene:
  - Branch **feature/hdf5** in Jülich repository.
  - Option in input file to activate HDF5 output.
  - Output of tallies.
  - Grid not fully implemented.
- Procedures required for HDF5 output are in module.
  - Easy to apply to other output files.

```
HDF5 "tallies.hdf5" {
FILE_CONTENTS {
 group      /
 group      /grid
   group      /grid/additionalGrid
   group      /grid/standardGrid
     dataset    /grid/standardGrid/nodes
 group      /intal
   dataset    /intal/001
   ...
   dataset    /intal/024
 group      /outtal
   group      /outtal/001
     dataset    /outtal/001/00
     ...
     dataset    /outtal/001/03
   ...
   group      /outtal/084
     dataset    /outtal/084/00
     ...
     dataset    /outtal/084/03
 group      /srftal
   dataset    /srftal/001
   ...
   dataset    /srftal/084
 }
}
```

Fig 1. Structure of HDF5 output in Eirene.

Object Oriented Programming

OOP philosophy

- In a OOP code, regardless of the language used, the basic unit to perform tasks becomes an *object*, which has its own variables and procedures.
- Objects can be extended, creating a hierarchy.
- The code is usually transparent to the object type. This means that the main program does not bother to check what type of object you are using, as it is responsibility of the object to perform tasks object dependent.

OOP simple example (I)

- Let us check how a very simple problem can be easily solved with OOP.
- An user inputs a series of geometries (circles, rectangles, triangles…) and wants to compute the area of all of them.
- First, a **TYPE** named *geometry* will be created. All other geometries will be extended from this type. The **TYPE** has a *calculateAera()* procedure which is **DEFERRED**, meaning that each extended type needs to implement it own procedure to calculate the area.

```
TYPE, PUBLIC, ABSTRACT:: geometry
  CONTAINS
    PROCEDURE(calculateArea_interface), PASS, DEFERRED:: calculateArea

END TYPE geometry

ABSTRACT INTERFACE
  FUNCTION calculateArea_interface(self) RESULT(area)
    IMPORT:: geometry
    CLASS(geometry), INTENT(in):: self
    REAL(8):: area

  END FUNCTION calculateArea_interface

END INTERFACE
```

OOP simple example (II)

- Now, each shape becomes an extension of *geometry* and has it own implementation of *calculateArea()*.
- Procedure needs to fulfill the interface defined in *geometry*.
- So, when the code wants to calculate the area of a specific geometry, it will call *calculateArea()* without knowing if the object is a rectangle or a circle.
- In a non-OOP approach, a manual way to identify geometries (usually an **INTEGER**), would have to be used and a **SELECT CASE** employed to calculate the specific area.
- This increases the code complexity and hinders its extension.

```
TYPE, PUBLIC, EXTENDS(geometry):: rectangle
  REAL(8):: l, h
  CONTAINS
    PROCEDURE, PASS:: calculateArea => calculateAreaRectangle

END TYPE rectangle

TYPE, PUBLIC, EXTENDS(geometry):: circle
  REAL(8):: r
  CONTAINS
    PROCEDURE, PASS:: calculateArea => calculateAreaCircle

END TYPE circle

FUNCTION calculateAreaRectangle(self) RESULT(area)
  IMPLICIT NONE

  CLASS(rectangle), INTENT(in):: self
  REAL(8):: area

  area = self%l * self%h

END FUNCTION calculateAreaRectangle

FUNCTION calculateAreaCircle(self) RESULT(area)
  USE constants, ONLY: PI
  IMPLICIT NONE

  CLASS(circle), INTENT(in):: self
  REAL(8):: area

  area = PI * self%r**2

END FUNCTION calculateAreaCircle
```

How basic OOP concepts are used for HDF5 and ASCII output

Simple OOP for HDF5 output in Eirene

- Currently, a simple implementation of OOP is used to deal with the output of tallies in ASCII and HDF5 formats.
- Eirene has different tallies: Input, Volume Averaged (Output) and Surface Averaged (Output).
- Each tally has different units and dimension and they are written in a different way.
- New abstract type for tallies, extended for each tally type.
- Each type has information about name, units, id and pointers to the data (same structure as before regarding data management).
- Each tally type has subroutines to write its own information in ASCII or HDF5 formats.
- Reduction of **IF** and **SELECT CASE** clauses.
- Much clearer code.
- Additional improvements could be done, but require a deeper modification of Eirene.

Examples of Code

```
26   TYPE, ABSTRACT :: tally
27     !id: Unique identification for tally
28     INTEGER:: id = 0
29     !name: Description of tally
30     !units: Units of the tally
31     CHARACTER(60):: name='FREEXX', units=' ---'
32     !active: indicates if the tally is active
33     LOGICAL, POINTER:: active => NULL()
34     CONTAINS
35       !Initialize an tally
36       PROCEDURE(initialize_interface), DEFERRED, PASS:: initialize
37       !Write the Tally as ASCII format
38       PROCEDURE(writeASCII_interface), DEFERRED, PASS:: writeASCII
39       !Write the Tally as HDF5 format
40       PROCEDURE(writeHDF5_interface),  DEFERRED, PASS:: writeHDF5
41
42   END TYPE tally
```

Fig 2. Generic type for tallies.

```
77   !Extension for input tally
78   TYPE, ABSTRACT, EXTENDS(tally):: tallyInput
79     !Type of header
80     INTEGER:: type = 0
81     CONTAINS
82       PROCEDURE, PASS:: initialize => initInputTally
83       PROCEDURE, PASS:: writeASCII => writeInputASCII
84       PROCEDURE, PASS:: writeHDF5  => writeInputHDF5
85       !Weighting of the tally
86       PROCEDURE, PASS:: weighting  => weightingInput
87       !Calculates the average value of the Tally
88       PROCEDURE, NOPASS:: average  => averageInput
89       !Integrate tally. Each extension needs to define its own integration
90       PROCEDURE(integrate_interface), DEFERRED, PASS:: integrate
91
92   END TYPE tallyInput
```

Fig 3. Extension for input tallies.

```
109  !Input tally with 1D data
110  TYPE, EXTENDS(tallyInput):: tallyInput1D
111    !Data of tally
112    REAL(DP), POINTER:: data(:)
113    CONTAINS
114      PROCEDURE, PASS :: integrate => integrate1D
115
116  END TYPE tallyInput1D
```

Fig 4. Input tally for 1D data.

```
118  !Input tally with 2D data
119  TYPE, EXTENDS(tallyInput):: tallyInput2D
120    !firstDimension: first dimension of the data array (firstdimension, number of cells)
121    INTEGER:: firstDImension=1
122    !Data of tally
123    REAL(DP), POINTER:: data(:,:)
124    CONTAINS
125      PROCEDURE, PASS :: integrate => integrate2D
126
127  END TYPE tallyInput2D
```

Fig 5. Input tally for 2D data.



Thank you for your attention

J. Gonzalez | TSVV-5 VC

DIFFER